Probabilidade e Estatística, 2010/2

CCT - UDESC

Prof. Fernando Deeke Sasse

Testes de Hipóteses para médias

1. A temperatura média da água descartada por uma torre de resfriamento não deve ser maior que 100F. A experiência indica que o desvio padrão da temperatura é 2F. A temperatura da água é medida em 9 dias aletóriamente escolhidos e a temperatura média encontrada é 98F.
(a) A temperatura da água é aceitável com $\alpha = 0.05$?
(b) Qual é o valor-P para este teste?
(c) Qual é a probabilidade de se aceitar a hipótese nula com $\alpha = 0.05$ se a água tem temperatura média verdadeira de 102F?
Solução:
Estabeleçamos a média 100F como um limite não tolerado:
$H_0 : \mu < 100, \quad H_1 : \mu \geq 100$
(a)

```
> restart :
> with(Statistics) :
> μ := 100; σ := 2; n := 9; α := 0.05;
```

$$\mu := 100$$
$$\sigma := 2$$
$$n := 9$$
$$\alpha := 0.05 \qquad (1.1)$$

```
> X := RandomVariable(Normal(μ, σ/sqrt(n))) :
> xc := Quantile(X, α)
```

$$xc := 98.90343092 \qquad (1.2)$$

O intervalo de rejeição é $(98.90, \infty)$. Como a média observada 98F está fora deste intervalo, a hipótese nula é consistente com a observação.
(b)

```
> xobs := 98 :
> evalf(CDF(X, xobs))
```

$$0.0013498980 \qquad (1.3)$$

(c)

```
> X := RandomVariable(Normal(102, σ/sqrt(n))) :
> β := evalf(CDF(X, xc))
```

$$\beta := 0.000001701588324 \qquad (1.4)$$

Ou seja, $\beta = 0$.

2. Uma fábrica produz virabrequins para motores de carro. O desgaste do virabrequim (em unidades de 0.0001 in) depois de 100,000 milhas é de interesse, pois ele tem impacto nos consertos dentro da

garantia. Uma amostra aleatória de 15 peças é testada e o desgaste médio é x_{obs} =2.78. É sabido que σ =0.9 e que o desgaste é normalmente distribuído.
(a) Teste a hipótese H_0: $\mu = 3$ contra H_1: $\mu \neq 3$, usando α=0.05.
(b) Qual é o poder do teste se $\mu = 3.25$
(c) Qual deve ser o tamanho da amostra para detectar a verdadeira média de 3.75 se o poder do teste deve ser ao menos 0.9?
Solução:
(a)

> $restart:$

> $with(Statistics):$

> $\mu := 3; \sigma := 0.9; n := 15; \alpha := 0.05; xobs := 2.78$

$$\mu := 3$$
$$\sigma := 0.9$$
$$n := 15$$
$$\alpha := 0.05$$
$$xobs := 2.78 \qquad \textbf{(2.1)}$$

> $X := RandomVariable\left(Normal\left(\mu, \frac{\sigma}{\text{sqrt}(n)}\right)\right):$

> $x1 := Quantile\left(X, \frac{\alpha}{2}\right)$

$$x1 := 2.544545528 \qquad \textbf{(2.2)}$$

> $x2 := Quantile\left(X, 1 - \frac{\alpha}{2}\right)$

$$x2 := 3.455454472 \qquad \textbf{(2.3)}$$

Como o valor observado está dentro do intervalo de aceitação $(x1, x2)$, a hipótese nula é aceita. Este problema pode também ser resolvido calculando o intervalo de confiança para x_{obs} e verificando que $\mu = 3$ está contido neste intervalo (exercício).
(b)

> $X := RandomVariable\left(Normal\left(3.25, \frac{\sigma}{\text{sqrt}(n)}\right)\right):$

> $\beta := CDF(X, x2) - CDF(X, x1)$

$$\beta := 0.8104888825 \qquad \textbf{(2.4)}$$

Portanto, o poder do teste é

> $1 - \beta$

$$0.1895111175 \qquad \textbf{(2.5)}$$

(c)

> $\delta := 3.75 - 3; \beta := 0.1$

$$\delta := 0.75$$
$$\beta := 0.1 \qquad \textbf{(2.6)}$$

> $Z := RandomVariable(Normal(0, 1)):$

> $z_{\frac{\alpha}{2}} := Quantile\left(Z, 1 - \frac{\alpha}{2}\right)$

$$z_{0.02500000000} := 1.959963985 \quad \textbf{(2.7)}$$

> $z_{\beta} := Quantile(Z, 1 - \beta)$

$$z_{0.1} := 1.281551566 \quad \textbf{(2.8)}$$

> $N := \dfrac{\left(z_{\frac{\alpha}{2}} + z_{\beta}\right)^2 \cdot \sigma^2}{\delta^2}$

$$N := 15.13068922 \quad \textbf{(2.9)}$$

Ou seja, ao menos 16 amostras.

3. Um artigo no Journal of Composite Materials, (December 1989, Vol 23, p. 1200) descreve o efeito de delaminação na frequência natural de barras feitas a partir de laminados compostos. 5 barras delaminadas são sujeitas a cargas, e as frequências resultantes são as seguintes (Hz): 230.66, 233.05, 232.58, 229.48, 232.58. Determine um intervalo de confiança de 90% sobre a média. Há evidências que suportem a suposição de normalidade da população? Teste a hipótese de que a média é 233.
Solução:

> $restart$:

> $with(Statistics)$:

> $L := [230.66, 233.05, 232.58, 229.48, 232.58]$:

> $NormalPlot(L)$



A distribuição amostral é indica uma distribuição aproximadamente normal da população.

> $n := 5; \text{alpha} := 0.1$

$$n := 5$$
$$\alpha := 0.1 \quad \textbf{(3.1)}$$

> $\mu := Mean(L);$

$$\mu := 231.6700000 \quad \textbf{(3.2)}$$

> $s := StandardDeviation(L)$

$$s := 1.53107805157021 \quad \textbf{(3.3)}$$

> $k := n-1$

$$k := 4 \tag{3.4}$$

> $T := RandomVariable(StudentT(k)) :$

> $tc := Quantile\left(T, 1 - \frac{\text{alpha}}{2}\right)$

$$tc := 2.131822837 \tag{3.5}$$

> $XC := evalf\left(\frac{tc \cdot s}{\text{sqrt}(n)}\right)$

$$XC := 1.45969943118047 \tag{3.6}$$

> $L1 := \mu - XC; L2 := \mu + XC$

$$L1 := 230.210300568820$$
$$L2 := 233.129699431180 \tag{3.7}$$

Portanto, o intervalo é [230.21, 233.13]. Como o valor 233 associado à hipótese nula está dentro deste intervalo, a hipótese nula é aceita.

4. Todos os cigarros atualmente vendidos no mercado têm um conteúdo médio de nicotina de ao menos 1.6 mg por cigarro. Uma empresa que produz cigarros afirma ter descoberto um novo processo que resulta em cigarros com menos de 1.6g de nicotina, em média. Para testar esta afirmação 20 amostras deste tipo de cigarro foram analisadas. Se é sabido que o desvio padrão do conteúdo de nicotina de um cigarro é 0.8mg, quais conclusões podem ser obtidas, com um nível de significância de 5%, se o conteúdo médio de nicotina de 20 cigarros é 1.54mg?

Solução:

Necessitamos inicialmente decidir sobre a hipótese nula apropriada. A abordagem para o teste não é simétrica com relação às hipóteses nula e alternativa, pois consideramos somente testes com a propriedade de que a probabilidade de rejeitar a hipótese nula quando ela é verdadeira (erro tipo I) nunca excedo o nível de significância α. Portanto, enquanto que a rejeição da hipótese nula é uma afirmação forte sobre os dados não serem consistentes com esta hipótese, uma afirmação análoga não pode ser feita quando a hipótese nula é aceita. Portanto, vamos endossar a afirmação do produtor somente quando houver evidência substancial para isso. Tal afirmação será a hipótese alternativa. Devemos então testar:

$H_0 : \mu \geq 1.6$ versus $H_1 : \mu < 1.6$.

> $with(Statistics) :$

> $n := 20 : xobs := 1.54 : \mu := 1.6 : \alpha := 0.05 : \sigma := 0.8 :$

> $X := RandomVariable\left(Normal\left(\mu, \frac{\sigma}{\text{sqrt}(n)}\right)\right) :$

> $xc := Quantile(X, \alpha)$

$$xc := 1.305759638 \tag{4.1}$$

Ou seja, o intervalo de rejeição é (1.306, ∞). Como 1.54 está dentro deste intervalo, não podemos rejeitar a hipótese nula, ou seja, $\mu \geq 16$ com um nível de significância de 0.05. Ou seja, embora os dados não invalidem o afirmação do fabricante, eles não são suficientemente fortes para corroborá-las. A mesmo conclusão poderia ter sido obtida verificando-se o valor P do teste:

> $CDF(X, \text{xobs})$

$$0.368657838608208777 \tag{4.2}$$

Tal valor é consideravelmente maior que 0.05.

5. O fornecedor de água de uma cidade afirma que consumo médio de água em residências é de 350 galões por dia. Para verificar tal afirmação, 20 casas foram aleatoriamente selecionadas e os seguintes valores de consumo foram obtidos:
340, 344, 362, 375, 356, 386, 354, 364, 332, 402, 340, 355, 362, 322, 372, 324, 318, 360, 338, 370.
Estes dados contradizem a afirmação oficial?
Solução:
Devemos testar as hipóteses:
$H_0 : \mu = 350$ versus $H_1 : \mu \neq 350$

```
> restart :
> with(Statistics) :
> μ := 350 :
> L := [340, 344, 362, 375, 356, 386, 354, 364, 332, 402, 340, 355, 362, 322, 372, 324,
     318, 360, 338, 370] :
> xobs := Mean(L)
```

$$xobs := 353.8000000 \quad (5.1)$$

```
> s := StandardDeviation(L)
```

$$s := 21.8477988774493 \quad (5.2)$$

```
> n := nops(L)
```

$$n := 20 \quad (5.3)$$

```
> NormalPlot(L)
```




```
> k := n − 1; alpha := 0.05
```

$$k := 19$$

$$\alpha := 0.05 \quad (5.4)$$

```
> T := RandomVariable(StudentT(k)) :
```

$$> tc := Quantile\left(T, 1 - \frac{alpha}{2}\right);$$

$$tc := 2.093024048 \quad (5.5)$$

$$> XC := evalf\left(\frac{tc \cdot s}{\text{sqrt}(n)}\right)$$

$$XC := 10.2250845896192 \quad (5.6)$$

> $L1 := \mu - XC;\ L2 := \mu + XC$

$$L1 := 339.774915410381$$
$$L2 := 360.225084589619 \quad (5.7)$$

Como a média observada 353.80 está dentro deste intervalo, a hipótese nula é aceita com um nível de significância de 5%. Portanto, os dados não são inconsistentes com a afirmação oficial.
De fato, como

> $tobs := evalf\left(\dfrac{(xobs - \mu)\cdot \mathrm{sqrt}(n)}{s}\right)$

$$tobs := 0.777841132670847 \quad (5.8)$$

o valor-P para este teste é

> $P := 2\cdot(1 - evalf(CDF(\mathrm{T}, \mathrm{tobs})))$

$$P := 0.446241090 \quad (5.9)$$

o que indica também que xobs está dentro do intervalo de aceitação.

6. Supercavitação é um fenômeno estudado na tecnologia de propulsão submarina a altas velocidades. Ela ocorre a velocidades acima de aproximadamente 50m/s, quando a pressão cai o suficiente para permitir que a água de vaporize, formando uma bolha de gás atrás do veículo. Quando a bolha de gás envolve completamente o veículo, supercavitação ocorre. 8 testes são realizados com modelos em escala reduzida e uma velocidade média de 102.2m/s é observada. Suponha que a velocidade é normalmente distribuída e $\sigma = 4$m/s.
(a) Teste a hipótese $H_0 : \mu = 100$ versus $H_0 : \mu \neq 100$, usando $\alpha = 0.05$.
(b) Calcule o poder do teste se a velocidade média verdadeira é tão baixa quanto 95m/s.
(c) Qual o tamanho da amostra requerido para detectar uma média verdadeira tão baixa quanto 95m/s, se queremos que o poder do teste seja de ao menos 0.85?
Solução:
(a)
$H_0 : \mu = 100$ versus $H_1 : \mu \neq 100$.

> $restart$

> $with(Statistics)$:

> $n := 8 : xobs := 102.2 : \mu := 100 : \alpha := 0.05 : \sigma := 4 :$

> $X := RandomVariable\left(Normal\left(\mu, \dfrac{\sigma}{\mathrm{sqrt}(n)}\right)\right) :$

> $x1 := Quantile\left(X, \dfrac{\mathrm{alpha}}{2}\right)$

$$x1 := 97.22819235 \quad (6.1)$$

> $x2 := Quantile\left(X, 1 - \dfrac{\mathrm{alpha}}{2}\right)$

$$x2 := 102.7718076 \quad (6.2)$$

O intervalo de aceitação é, portanto, [97.228,102.771]. Como a média observada está no interior deste intervalo a hipótese nula é aceita.

(b)

> $X := RandomVariable\left(Normal\left(95, \dfrac{\sigma}{\mathrm{sqrt}(n)}\right)\right) :$

> $\beta := CDF(X, x2) - CDF(X, x1)$

$$\beta := 0.0575624569 \quad \textbf{(6.3)}$$

O poder do teste é então:

> $1 - \beta$

$$0.9424375431 \quad \textbf{(6.4)}$$

(c)

> $\delta := 100 - 95;$ beta $:= 0.15$

$$\delta := 5$$
$$\beta := 0.15 \quad \textbf{(6.5)}$$

> $Z := RandomVariable(Normal(0, 1)):$

> $z_{\frac{\alpha}{2}} := Quantile\left(Z, 1 - \frac{\alpha}{2}\right)$

$$z_{0.02500000000} := 1.959963985 \quad \textbf{(6.6)}$$

> $z_{\beta} := Quantile(Z, 1 - \beta)$

$$z_{0.15} := 1.036433389 \quad \textbf{(6.7)}$$

> $N := \frac{\left(z_{\frac{\alpha}{2}} + z_{\beta}\right)^2 \cdot \sigma^2}{\delta^2}$

$$N := 5.746174224 \quad \textbf{(6.8)}$$

Ou seja, ao menos 6 amostras são necessárias.

7. Um fabricante está formulando um novo shampoo e está interessada na altura da espuma (em mm). A altura da espuma é normalmente distribuída e tem desvio-padrão de 20mm. O fabricante deseja testar $H_0 : \mu = 175\ mm$ contra $H_1 : \mu > 175\ mm$, usando os resultados de 10 amostras.

(a) Determine a probabilidade de erro do tipo I α se a região crítica é $x > 185\ mm$.
(b) Qual é a probabilidade de erro do tipo II se a média verdadeira é de $195 mm$?
Solução

> $restart:$

> $with(Statistics):$

> $\mu := 175 : \sigma := 20 : n := 10 : xc := 185 :$

> $X := RandomVariable\left(Normal\left(\mu, \frac{\sigma}{\mathrm{sqrt}(n)}\right)\right):$

> alpha $:= evalf(1 - CDF(X, xc))$

$$\alpha := 0.0569231491 \quad \textbf{(7.1)}$$

(b)

> $X2 := RandomVariable\left(Normal\left(195, \frac{\sigma}{\mathrm{sqrt}(n)}\right)\right):$

> $\beta := evalf(CDF(X2, 185))$

$$\beta := 0.0569231491 \quad \textbf{(7.2)}$$

8. No problema anterior suponha que os dados amostrais resultam em $x_{obs} = 190mm$.

(a) Qual é a sua conclusão?
(b) Quão pouco usual é esta média amostral se a verdadeira média é 175*mm*? Ou seja, qual é a probabilidade de observamos uma média amostral tão grande quanto 190*mm* (ou maior), se a média verdadeira é 175*mm*?

Solução
(a) O valor observado de 190mm está fora do intervalo de aceitação de H_0: (-∞,185). Portanto, isso implicaria na rejeição de H_0.
(b)

$> restart$

$> with(Statistics):$

$> \sigma := 20 : n := 10 : x_{obs} := 190 :$

$> X2 := RandomVariable\left(Normal\left(175, \frac{\sigma}{\mathrm{sqrt}(n)}\right)\right):$

$> evalf(1 - CDF(X2, 190))$

$$0.0088530329 \quad \textbf{(8.1)}$$

9. Repita o problema 7 supondo que o tamanho da amostra da amostra é 16 e que a região crítica é a mesma.

Solução
(a)

$> restart:$

$> with(Statistics):$

$> \mu := 175 : \sigma := 20 : n := 18 : xc := 185 :$

$> X := RandomVariable\left(Normal\left(\mu, \frac{\sigma}{\mathrm{sqrt}(n)}\right)\right):$

$> \mathrm{alpha} := evalf(1 - CDF(X, xc))$

$$\alpha := 0.0169474268 \quad \textbf{(9.1)}$$

(b)

$> X2 := RandomVariable\left(Normal\left(195, \frac{\sigma}{\mathrm{sqrt}(n)}\right)\right):$

$> \beta := evalf(CDF(X2, 185))$

$$\beta := 0.0169474268 \quad \textbf{(9.2)}$$

10. Considere o problema 7 e suponha que o tamanho da amostra é 16.

(a) Qual deve ser a região crítica se a probabilidade de erro do tipo I deve permanecer a mesma que no caso em que $n = 10$?
(b) Usando a região crítica encontrada em (a), determine a probabilidade de erro do tipo II se a média

verdadeira é 195*mm*.

(c) Comparando o valor de β obtido em (b) com aquele do problema 7 quais conclusões podem ser obtidas?

Solução

(a)

```
> restart :
> with(Statistics) :
> μ := 175 : σ := 20 : n := 16 : α := 0.0569231491 :
> X := RandomVariable(Normal(μ, σ/sqrt(n))) :
> xc := Quantile(X, 1 - alpha)
```

$$xc := 182.9056941 \quad \textbf{(10.1)}$$

(b)

```
> X2 := RandomVariable(Normal(195, σ/sqrt(n))) :
> β := evalf(CDF(X2, xc))
```

$$\beta := 0.007784590298 \quad \textbf{(10.2)}$$

(c) A probabilidade de se aceitar a hipótese nula, $H_0 : \mu = 175$ *mm* dado que ela é falsa, ou seja, $H_1 : \mu \geq 182.90$ *mm*, diminui drasticamente, para um α fixo, quando o número de amostras aumenta de 10 para 16.